Maple-Befehle

Autor: S. Hussmann

1. Allgemeines und Konstanten

;	Abschluss einer Eingabe zur Ausführung mit <Enter>
:	wie oben, das Ergebnis wird aber nicht angezeigt
%,%%	Zugriff auf das letzte Ergebnis, vorletzte Ergebnis
#	Einleitung eines Kommentars
?	Aktivierung einer Online-Hilfe (z.B. ?evalf)
:=	Operator für die Zuweisung von Objekten an Variablen
x:='x';	Löschen der Variablenzuweisung von x
with(package)	Laden eines Paketes package mit Zusatzprogrammen
restart	Neustart, der alle bisherigen Zuweisungen löscht
Digits	legt die Anzahl Stellen für numerische Werte fest
print(expr)	führt zur Anzeige des Ausdrucks expr
Pi	die Zahl Π=3.14159...
exp(1)	die Eulersche Zahl e=2.718281...
I	die imaginäre Einheit einer komplexen Zahl z
infinity	Unendlich (quergestelltes 8)

2. Exakte Arithmetik und Gleitpunktarithmetik

+, - ,*, /	Grundrechenarten
^ oder **	Potenz
!	Fakultät
sqrt(x),root(x,n)	Quadratwurzel aus x, n-te Wurzel aus x
exp(x)	Exponentialfunktion von x
log(x), ln(x)	Natürlicher Logarithmus von x
sin(x),cos(x),tan(x),cot(x)	trigonometrische Funktionen von x
abs(x)	Betrag einer Zahl x
evalf(expr)	numerische Auswertung eines Ausdrucks expr
evalf(expr,n)	numerische Auswertung von expr auf n Stellen

Beispiele:

```
> # evalf angewendet auf Pi/3 auf 20 Stellen genau
> sin(Pi/3);
```

$$\frac{\sqrt{3}}{2}$$

```
> evalf(%,20);
```

$$0.86602540378443864675$$

```
>
>
> # gewünscht ist der dekatische Logarithmus von 10:
> log10(10);
```

$$1$$

```
>
```

3. Polynome und rationale Terme

collect(p,x)	fasst alle Koeffizienten im Polynom p mit gleicher Potenz von x zusammen
sort(p)	sortiert Ausdruck(Polynom) p nach Potenzen von x
factor(p)	zerlegt das Polynom p (falls möglich) in Faktoren
coeff(p,x,n)	ermittelt im Polynom p den zur n-ten Potenz von x gehörenden Koeffizienten
expand(expr)	multipliziert die in expr enthaltenen Faktoren aus
simplify(expr)	versucht, den Ausdruck expr zu vereinfachen
simplify(expr,option)	simplify mit den Optionen: power, trig, sqrt, radical
radsimp(expr)	vereinfacht einen Ausdruck expr, der Wurzeln enthält
numer(expr)	ermittelt den Zähler des rationalen Ausdruckes expr
denom(expr)	ermittelt den Nenner des rationalen Ausdruckes expr
subs(x=a,expr)	substituiert im Ausdruck expr die Varable x durch a. a kann eine Zahl oder ein Ausdruck sein.
subs(var1=rep1,var2=rep2,expr)	Substituiert in expr. var1 durch rep1 und var2 durch rep2
combine(expr)	lässt in einem Ausdruck expr Terme verschmelzen
combine(expr,option)	combine mit den Optionen: exp, trig, ln, power
normal	macht Terme Gleichnennerig (Zähler und Nenner faktrorisiert)
convert(expr,rational,n)	Approximation von expr durch eine rationale Zahl als Bruch in einer von n abhängigen Genauigkeit
convert(expr,option)	convert mit den Optionen: exp, ln, tan, trig, expln, expsincos, sincos, parfrac, binary, rational

Beispiele:

```
> # expand
> (x+y)*(x-y)-x^2;
```

$$(x+y)(x-y)-x^2$$

```
> expand(%);
```

$$-y^2$$

```
>
>
> # factor
> x^4+x^2*y+2*x^2+2*x^3+2*x*y+2*x+y+1;
```

$$x^4+x^2\,y+2\,x^2+2\,x^3+2\,x\,y+2\,x+y+1$$

```
> factor(%);
```

$$(x+1)^2\,(x^2+y+1)$$

```
>
>
> # simplify
> simplify(exp(a+ln(b*exp(c))));
```

$$b\,\mathrm{e}^{(a+c)}$$

```
>
>
> # combine mit der Option trig
> e1:=2*cos(x/2)^2-1;
```

$$e1 := 2\cos\left(\frac{x}{2}\right)^2-1$$

```
> e2:=cos(x)^2+2*sin(x)^2+tan(x)^2;
```

$$e2 := \cos(x)^2+2\sin(x)^2+\tan(x)^2$$

```
> combine(e2,trig);
```

$$-\frac{1}{2}\cos(2\,x)+\frac{3}{2}+\tan(x)^2$$

```
>
>
> # combine mit den Optionen exp, ln, power
> e3:=(exp(x))^2*exp(y)+2*ln(x)-ln(y)+(x^a)^2;
```

$$e3 := (\mathrm{e}^x)^2\,\mathrm{e}^y+2\ln(x)-\ln(y)+(x^a)^2$$

```
> combine(e3,exp);
```

$$\mathrm{e}^{(2\,x+y)}+2\ln(x)-\ln(y)+(x^a)^2$$

```
> combine(e3,ln);
```

$$(\mathrm{e}^x)^2\,\mathrm{e}^y+2\ln(x)-\ln(y)+(x^a)^2$$

```
> combine(e3,power);
```

$$\mathrm{e}^{(2\,x+y)}+2\ln(x)-\ln(y)+x^{(2\,a)}$$

```
> combine(e3,trig); # macht keinen Sinn
```

$$(\mathrm{e}^x)^2\,\mathrm{e}^y+2\ln(x)-\ln(y)+(x^a)^2$$

```
>
>
> # simplify
> expr:=cos(a)^2*cos(b)^2-
2*cos(a)*cos(b)*sin(a)*sin(b)+sin(a)^2*sin(b)^2;
```

$$expr := \cos(a)^2\cos(b)^2-2\cos(a)\cos(b)\sin(a)\sin(b)+\sin(a)^2\sin(b)^2$$

```
> simplify(expr);
```

$$2\cos(a)^2\cos(b)^2-2\cos(a)\cos(b)\sin(a)\sin(b)+1-\cos(b)^2-\cos(a)^2$$

```
> # Bringt fast nichts!
> combine(expr,trig);
```

$$\frac{1}{2}\cos(2\,a+2\,b)+\frac{1}{2}$$

```
> convert(%,tan);
```

$$\frac{1}{2}\,\frac{1-\tan(a+b)^2}{1+\tan(a+b)^2}+\frac{1}{2}$$

```
> simplify(%);
```

$$\frac{1}{1+\tan(a+b)^2}$$

```
> convert(%,cos);
```

$$\frac{1}{1+\dfrac{\cos\left(a+b+\dfrac{\pi}{2}\right)^2}{\cos(a+b)^2}}$$

```
> simplify(%);
```

$$\cos(a+b)^2$$

```
>
```

```
>
> # convert mit einer Binärzahl
> convert(51001,binary);
```

$$1100011100111001$$

```
> convert(51001,binary,30);
```

$$1100011100111001$$

```
>
>
> # Partialbruchzerlegung mit convert und der Option parfrac
> convert(2/(x^2-1),parfrac,x);
```

$$\frac{1}{x-1}-\frac{1}{x+1}$$

```
> convert((x^3+2*x^2-x)/(x^3-x^2+x-1),parfrac,x);
```

$$1+\frac{1}{x-1}+\frac{2\,x}{x^2+1}$$

```
>
>
> # normal
> e4:=x/y-x/(x+y);
```

$$e4:=\frac{x}{y}-\frac{x}{x+y}$$

```
> normal(e4,expanded);
```

$$\frac{x^2}{x\,y+y^2}$$

```
>
>
> # Substitution
> soln:=solve(3*x-y=x,x);
```

$$soln:=\frac{y}{2}$$

```
> subs(x=soln,3*x-y=x);
```

$$\frac{y}{2}=\frac{y}{2}$$

```
> f:=x*y^2+z;
```

$$f:=x\,y^2+z$$

```
> subs(x=y,y=x,f); # nacheinander!
```

$$x^3+z$$

```
> subs({x=y,y=x},f); # simultan!
```

$$y\,x^2+z$$

```
>
```

4. Befehle aus der Analysis

diff(f(x),x)	partielle Ableitung
int(f(x),x)	unbestimmte Integration
sum(f(x),x)	unbestimmte Summation
int(f(x),x=a..b)	bestimmte Integration
sum(f(x),x=a..b)	bestimmte Summation
limit(f(x),x=a,option)	Grenzwerte (limus der Funktion f(x) und x geht gegen a)

Beispiele:

```
> # Differentieren
> diff(ln(x),x);
```

$$\frac{1}{x}$$

```
> diff(arcsin(a*x),x);
>
```

$$\frac{a}{\sqrt{1-a^2\,x^2}}$$

```
>
>
> # Integrieren
> int(sqrt(1-x^2),x);
```

$$\frac{x\sqrt{1-x^2}}{2}+\frac{1}{2}\arcsin(x)$$

```
> int(ln(x),x=1..2);
```

$$-1+2\ln(2)$$

```
>
>
> # Unbestimmte Summation
> sum(n,n);
```

$$\frac{1}{2}n^2-\frac{1}{2}n$$

```
> F:=n->1/2*n^2-1/2*n;
```

$$F := n \rightarrow \frac{1}{2}n^2-\frac{1}{2}n$$

```
> sum(n,n=1..4);
```

$$10$$

```
> F(5)-F(1);
```

$$10$$

```
>
>
> # Grenzwerte
> limit(sin(x)/x,x=0);
```

$$1$$

```
> limit(tan(x),x=Pi/2);
```

$$\mathit{undefined}$$

```
> limit(tan(x),x=Pi/2,right);
```

$$-\infty$$

```
> limit(tan(x),x=Pi/2,left);
```

$$\infty$$

```
>
```

5. Gleichungen und Ungleichungen

lhs(eq), rhs(eq)	liefert linke bzw. rechte Seite der Gleichung eq
allvalues(s)	ermittelt numerisch alle Lösungen vom RootOf-Ausdruck s
solve(eq,x)	löst die Gleichung/Ungleichung eq nach der Variablen x auf
solve({e1,…,en}, {x1,…,xm})	löst die Gleichungen e1,…,en nach den Variablen x1,…,xm auf
fsolve(e,x)	ermittelt numerisch eine Lösung x
fsolve(e,x,x=a..b)	ermittelt numerisch eine Lösung im Intervall [a,b]
fsolve(e,x,complex)	bringt auch komplexe Lösungen von e
solve({e1,…,en}, {x1,…,xm})	löst numerisch die Gleichungen e1,…,en nach den Variablen x1,…,xm auf
rsolve(e, f(n))	löst die Rekursionsgleichung e nach f(n) auf
rsolve({e, i1,…,in}, f(n))	löst die Rekursionsgleichung e mit den Anfangsbedingungen i1,…,in nach f(n) auf
dsolve(e, y(x))	löst die DGL e nach y(x) auf
dsolve({e, i1,…,in}, y(x))	löst die DGL e mit den Anfangsbedingungen i1,…,in nach y(x) auf
dsolve({e1,…,en, i1,…,in}, {y1(x),…,yn(x)}, laplace)	Lösung mit Hilfe der LT (vor allem für DGL mit Anfangsbedingungen und für DGL-Systeme geeignet)
dsolve({e1,…,en, i1,…,in}, {y1(x),…,yn(x)}, series)	Lösung durch Potenzreihenentwicklung der gesuchten Funktion (bis zum Glied O(x^6))
dsolve({e1,…,en, i1,…,in}, {y1(x),…,yn(x)}, explicit)	Explizite Lösung der DGL
dsolve({e1,…,en, i1,…,in}, {y1(x),…,yn(x)}, numeric)	numerische Lösung der DGL (besonders bei Anfangswertproblemen geeignet)

Bemerkungen zu Anfangsbedingungen:

Mathematisch: $y^{(n)}(0) = y_0$
Maple Notation: (D@@n)(y)(0)=y0

y'(0)=5 <-> D(y)(0)=5
y''(0)=3 <-> (D@@2)(y)(0)=3
y'''(0)=4 <-> (D@@3)(y)(0)=4
oder
y''(0)=3 <-> D(D(y))(0)=3
y'''(0)=4 <-> D(D(D(y)))(0)=4

Beispiele:

```
> # solve
> fsolve({y^2+1=x,x+2=y},{x,y},complex);
```

$$\{x = -1.500000000 - 1.658312395\, I,\ y = 0.5000000000 - 1.658312395\, I\}$$

```
> eqnset:={x+y=b,a*x-2/3*y=k};
```

$$eqnset := \{x + y = b,\ a\,x - \frac{2\,y}{3} = k\}$$

```
> varset:={x,y};
```

$$varset := \{x, y\}$$

```
> solutionset:=solve(eqnset,varset);
```

$$solutionset := \{y = \frac{3\,(a\,b - k)}{3\,a + 2},\ x = \frac{3\,k + 2\,b}{3\,a + 2}\}$$

```
>
>
> # fsolve
> e:=x^3+5*x^2-7*x+15;
```

$$e := x^3 + 5x^2 - 7x + 15$$

```
> solve(e,x); # zu kompliziert!
```

$$-\frac{(485+3\sqrt{15321})^{(1/3)}}{3} - \frac{46}{3(485+3\sqrt{15321})^{(1/3)}} - \frac{5}{3}, \frac{(485+3\sqrt{15321})^{(1/3)}}{6} + \frac{23}{3(485+3\sqrt{15321})^{(1/3)}} - \frac{5}{3} + \frac{1}{2} I\sqrt{3}\left(-\frac{(485+3\sqrt{15321})^{(1/3)}}{3} + \frac{46}{3(485+3\sqrt{15321})^{(1/3)}}\right),$$
$$\frac{(485+3\sqrt{15321})^{(1/3)}}{6} + \frac{23}{3(485+3\sqrt{15321})^{(1/3)}} - \frac{5}{3} - \frac{1}{2} I\sqrt{3}\left(-\frac{(485+3\sqrt{15321})^{(1/3)}}{3} + \frac{46}{3(485+3\sqrt{15321})^{(1/3)}}\right)$$

```
> fsolve(e,x); # bringt nur die relle Lösung
```

$$-6.446740094$$

```
> fsolve(e,x,complex); # alle Lösungen
```

$$-6.446740094, 0.7233700468 - 1.342941973\, I, 0.7233700468 + 1.342941973\, I$$

```
>
>
> # Diffgleichungen 1
> dsolve(diff(f(x),x)+f(x)^5*x=sin(x),f(x));
> # kann von Maple nicht gelöst werden
> dsolve({diff(f(x),x)+f(x)^5*x=sin(x),f(0)=1/2},f(x),series);
```

$$f(x) = \frac{1}{2} + \frac{31}{64}x^2 - \frac{977}{12288}x^4 + O(x^6)$$

```
>
>
> # Diffgleichungen 2
>
dgl:=diff(y(x),x$3)+8*diff(y(x),x,x)+17*diff(y(x),x)+10*y(x)=4*exp(-x);
```

$$dgl := \left(\frac{d^3}{dx^3}y(x)\right) + 8\left(\frac{d^2}{dx^2}y(x)\right) + 17\left(\frac{d}{dx}y(x)\right) + 10\, y(x) = 4\, e^{(-x)}$$

```
> dsolve({dgl,y(0)=-1,D(y)(0)=1,(D@@2)(y)(0)=0},y(x));
```

$$y(x) = \frac{x}{e^x} - 2\, e^{(-x)} + e^{(-2x)}$$

```
> assign(%);
> plot(y(x),x=0..10);
```




```
> f:=unapply(y(x),x);
```

$$f := x \to \frac{x}{e^x} - 2\, e^{(-x)} + e^{(-2\, x)}$$

```
> f(0);
```

$$-1$$

```
> f(1);
```

$$\frac{1}{e} - 2\, e^{(-1)} + e^{(-2)}$$

```
> evalf(%);
```

$$-0.2325441580$$

```
> f(t);
```

$$\frac{t}{e^t} - 2\, e^{(-t)} + e^{(-2\, t)}$$

```
> f(omega*t+alpha);
```

$$\frac{\omega\, t + \alpha}{e^{(\omega\, t + \alpha)}} - 2\, e^{(-\omega\, t - \alpha)} + e^{(-2\, \omega\, t - 2\, \alpha)}$$

```
>
>
> # Diffgleichungen 3
> ?dsolve[numeric]
> ?DEplots
> f:=dsolve({diff(y(x),x)=x-y(x)^2,y(0)=-0.5},y(x),numeric);
```

$f :=$ proc (x_rkf45) ... end proc

```
> f(0); f(.2); f(.4);
```

$$[x = 0.,\ y(x) = -0.50000000000000]$$

$$[x = 0.2,\ y(x) = -0.534051783619025167]$$

$$[x = 0.4,\ y(x) = -0.531549144732234802]$$

```
> de:=diff(y(x),x)=(x-y(x))/(1+x^2+y(x)^2);
```

$$de := \frac{d}{dx}\,\mathrm{y}(x) = \frac{x - \mathrm{y}(x)}{1 + x^2 + \mathrm{y}(x)^2}$$

```
> ic:=y(0)=1;
```

$$ic := \mathrm{y}(0) = 1$$

```
> f1:=dsolve({de,ic},y(x),numeric);
```

$$f1 := \mathbf{proc}\,(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

```
> f1(0); f1(.4); f1(10);
```

$$[\,x = 0.,\ \mathrm{y}(x) = 1.\,]$$

$$[\,x = 0.4,\ \mathrm{y}(x) = 0.849358609361580252\,]$$

$$[\,x = 10.,\ \mathrm{y}(x) = 1.91493000584053474\,]$$

```
> ?odeplot
> with(plots):
```

Warning, the name changecoords has been redefined

```
> odeplot(f1,[x,y(x)],0..10);
```




```
>
```

6. Funktionen

->	Pfeiloperator, der eine Funktion f(x) von einer oder g(x,y,...) von mehreren Variablen definiert
unapply(expr(x,y,..),x,y,..)	wandelt den Ausdruck expr in den Variablen x,y,.. in eine Funktion um
erf(x)	Gausssches Fehlerintegral
binomial(n,k)	binomial Funktion
GAMMA(x)	Gamma Funktion
Psi(x)	Poly Gamma Funktion
Zeta(x)	Riemannsche Zeta Funktion
BesselJ(v,x)	Besselsche Funktion ersten Grades

Beispiele:

```
> # Definieren einer Funktion
> f:=x->2*sin(x);
```

$$f := x \rightarrow 2 \sin(x)$$

```
> g:=(x,y,z)->x+x*z-y;
```

$$g := (x, y, z) \rightarrow x + x\,z - y$$

```
>
>
> # Beispiel mit unapply
> g:=x^2+1;
```

$$g := x^2 + 1$$

```
> f:=unapply(g,x);
```

$$f := x \rightarrow x^2 + 1$$

```
>
```

7. Datenstrukturen

F:=a,b,c;	Folge (a,b,c können beliebige Ausdrücke sein)
L:=[F]	Umwandlung der Folge F in eine Liste L
M:={F}	Umwandlung einer Folge F in eine Menge M
op(L)	Umwandlung der Liste L in eine Folge
nops(L)	Anzahl der Elemente der Liste L
F[n], L[n]	n-tes Element der Folge F, der Liste L
F[i..j], L[i..j]	Teilfolge bzw. Teilliste, die aus den i-ten bis j-ten Elementen besteht
seq(f(i),i=2..4)	Folge f(2),f(3),f(4) zu einer (vorher definierten) Funktion f(x)
seq(f(i),i=[2,4,6])	Folge f(2),f(4),f(6)
a union b	a vereinigt b
a subset b	a Teilmenge von b
a intersect b	a schneidet b

Beispiele:

```
> S1:=1,5,3;
```

$$S1 := 1, 5, 3$$

```
> max(S1);
```

$$5$$

```
> S2:=seq(i^2,i=1..5);
```

$$S2 := 1, 4, 9, 16, 25$$

```
> S2[4];
```

$$16$$

```
> S1:={x,y,z,y};
```

$$S1 := \{x, y, z\}$$

```
> L1:=[x,y,z,y];
>
```

$$L1 := [x, y, z, y]$$

```
> L2:=[L1[4],L1[2..3]];
```

$$L2 := [y, [y, z]]$$

```
> L3:=[a,b,c,y,z,y];
```

$$L3 := [a, b, c, y, z, y]$$

```
> A:={0,1,2,3,4,5};
```

$$A := \{0, 1, 2, 3, 4, 5\}$$

```
> B:={1,3,5,7,9,11};
```

$$B := \{1, 3, 5, 7, 9, 11\}$$

```
> # Vereinigung und Durchschnitt von 2 Mengen:
> A union B;
```

$$\{0, 1, 2, 3, 4, 5, 7, 9, 11\}$$

```
> A intersect B;
```

$$\{1, 3, 5\}$$

```
> # Auch die Differenz von 2 Mengen kann berechnet werden:
> A minus B;
```

$$\{0, 2, 4\}$$

```
>
```

8. Vektoren

vector(n)	Definition eines Vektors mit n Komponenten
vector([liste])	Def. eines Vektors und Eingabe der Komp. als Liste
v[i]	i-te Komponente des Vektors v
norm(v,2)	Norm eines Vektors v
normalize(v)	der zu v normierte Vektor

Bemerkung:
Ein Vektor ist in Maple ein eindimensionales Array

9. Matrizen

with(linalg)	ladet das Paket für lineare Algebra
matrix(m,n)	Def. einer Matrix mit m Zeilen und n Spalten
matrix(m,n,[liste])	Def. einer Matrix vom Typ (m,n) und zeilenweise Eingabe der Matrixelemente als Liste
M[i,j]	Element der Matrix M an Pos. (i,j)
row(M,i), col(M,j)	i-te Zeile, j-te Spalte der Matrix M
rowdim(M), coldim(M)	Zeilen-, Spaltenanzahl der Matrix M
diag([liste])	Def. einer Diagonalmatrix und Eingabe der Diagonalelemente als Liste
convert(v,matrix)	Umwandlung des Vektors v in 1-spaltige Matrix
evalm(expr)	Auswertung von Ausdrücken mit Vektoren und Matrizen
det(M)	gibt die Determinante der Matrix M zurück
inverse(M)	invertiert die Matrix M
evalm(A+B)	A+B
evalm(A-B)	A-B
evalm(A^2)	AxA
multiply(A,B)	AxB
transpose(A)	transponiert die Matrix A
charpoly(A,lambda)	der Befehl des charakteristischen Polynoms mit der Matrix A und der Variablen lambda
eigenvals(A)	berechnet die Eigenwerte der Matrix A
eigenvects(A)	berechnet die Eigenvektroren der Matrix A

Bemerkung:
Eine Matrix repräsentiert ein zweidimensionales Array

Beispiele:

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected

> A:=array([[x-1,2,3],[0,x-2,2],[2,1,x-3]]);
```

$$A := \begin{bmatrix} x-1 & 2 & 3 \\ 0 & x-2 & 2 \\ 2 & 1 & x-3 \end{bmatrix}$$

```
> # oder einfacher:
> A:=matrix(3,3,[x-1,2,3,0,x-2,2,2,1,x-3]);
```

$$A := \begin{bmatrix} x-1 & 2 & 3 \\ 0 & x-2 & 2 \\ 2 & 1 & x-3 \end{bmatrix}$$

```
> det(A);
```

$$x^3 - 6x^2 + 3x + 16$$

```
> inverse(A);
```

$$\begin{bmatrix} \frac{x^2-5x+4}{x^3-6x^2+3x+16} & -\frac{2x-9}{x^3-6x^2+3x+16} & -\frac{-10+3x}{x^3-6x^2+3x+16} \\ \frac{4}{x^3-6x^2+3x+16} & \frac{x^2-4x-3}{x^3-6x^2+3x+16} & -\frac{2(x-1)}{x^3-6x^2+3x+16} \\ -\frac{2(x-2)}{x^3-6x^2+3x+16} & -\frac{x-5}{x^3-6x^2+3x+16} & \frac{x^2-3x+2}{x^3-6x^2+3x+16} \end{bmatrix}$$

```
>
>
> # Eigenwerte und Eignvektoren
```

```
> with(linalg):
> A:=matrix(3,3,[1,0,5,0,1,1,1,1,0]);
```

$$A := \begin{bmatrix} 1 & 0 & 5 \\ 0 & 1 & 1 \\ 1 & 1 & 0 \end{bmatrix}$$

```
> charpoly(A,x);
```

$$x^3 - 2x^2 + 6 - 5x$$

```
> solve(%);
```

$$1, -2, 3$$

```
> eigenvals(A);
```

$$1, -2, 3$$

```
> eigenvects(A);
```

$$[-2, 1, \{[5, 1, -3]\}], [3, 1, \{[5, 1, 2]\}], [1, 1, \{[-1, 1, 0]\}]$$

```
> A:=matrix(3,3,[0,-1,-1,0,1,0,1,1,2]);
```

$$A := \begin{bmatrix} 0 & -1 & -1 \\ 0 & 1 & 0 \\ 1 & 1 & 2 \end{bmatrix}$$

```
> charpoly(A,x);
```

$$x^3 - 3x^2 + 3x - 1$$

```
> solve(%);
```

$$1, 1, 1$$

```
> eigenvals(A);
```

$$1, 1, 1$$

```
> eigenvects(A);
```

$$[1, 3, \{[-1, 1, 0], [-1, 0, 1]\}]$$

```
> A:=matrix(2,2,[3,1,-2,1]);
```

$$A := \begin{bmatrix} 3 & 1 \\ -2 & 1 \end{bmatrix}$$

```
> eigenvals(A);
```

$$2 + I, 2 - I$$

```
> eigenvects(A);
```

$$\left[2 + I, 1, \left\{\left[\frac{-1}{2} - \frac{1}{2} I, 1\right]\right\}\right], \left[2 - I, 1, \left\{\left[\frac{-1}{2} + \frac{1}{2} I, 1\right]\right\}\right]$$

```
> charmat(A,lambda); # Berechnet die char. Matrix von A mit dem
Parameter lambda
```

$$\begin{bmatrix} \lambda - 3 & -1 \\ 2 & \lambda - 1 \end{bmatrix}$$

```
> A:=matrix(3,3,[a,0,5,1,1,1,-a,0,0]);
```

$$A := \begin{bmatrix} a & 0 & 5 \\ 1 & 1 & 1 \\ -a & 0 & 0 \end{bmatrix}$$

```
> charmat(A,lambda);
```

$$\begin{bmatrix} \lambda - a & 0 & -5 \\ -1 & \lambda - 1 & -1 \\ a & 0 & \lambda \end{bmatrix}$$

```
> det(%);
```

$$\lambda^3 - \lambda^2 - a\,\lambda^2 + 6\,a\,\lambda - 5\,a$$

```
> charpoly(A,lambda);
```

$$\lambda^3 - \lambda^2 - a\,\lambda^2 + 6\,a\,\lambda - 5\,a$$

```
> collect(%,lambda);
```

$$\lambda^3 + (-1 - a)\,\lambda^2 + 6\,a\,\lambda - 5\,a$$

```
> solve(%,lambda);
```

$$1, \frac{a}{2} + \frac{\sqrt{a^2 - 20\,a}}{2}, \frac{a}{2} - \frac{\sqrt{a^2 - 20\,a}}{2}$$

```
> eigenvals(A);
```

$$1, \frac{a}{2} + \frac{\sqrt{a^2 - 20\,a}}{2}, \frac{a}{2} - \frac{\sqrt{a^2 - 20\,a}}{2}$$

```
>
>
> # noch etwas zu den Eigenvektoren:
> A:=matrix(3,3,[20.8,-44,111,12,-25.5,64,1,-2,5]);
```

$$A := \begin{bmatrix} 20.8 & -44 & 111 \\ 12 & -25.5 & 64 \\ 1 & -2 & 5 \end{bmatrix}$$

```
> eigen:=eigenvects(A);
```

$eigen :=$ [-2.782411948, 1, {[-25.35855012, -15.27248381, -0.6664280318]}],
[-0.1015892773, 1, {[-3.603749405, 6.374118830, 3.205273142]}],
[3.184001207, 1, {[-8.662025824, -4.816327875, -0.5344881977]}]

```
> # das Ergebnis erscheint in Form einer Liste. Jeder Eintrag in der
Liste ist wiederum ein geordnetes Tripel bestehend aus
> # (1) dem Eigenwert
> # (2) seiner Vielfachheit
> # (3) dem zugehörigen Eigenraum
>
> eintrag:=eigen[2];
```

$eintrag :=$ [-0.1015892773, 1, {[-3.603749405, 6.374118830, 3.205273142]}]

```
> # der Eigenwert ist der 1. Operand von eintrag:
> wert:=op(1,eintrag);
```

$wert :=$ -0.1015892773

```
> # der Eigenvektor ist im dritten Operanden "verpackt"
> op(3,eintrag);
```

{[-3.603749405, 6.374118830, 3.205273142]}

```
> vektor:=op(1,%);
```

$vektor :=$ [-3.603749405, 6.374118830, 3.205273142]

```
>
```

10. Grafikbefehle

with(plots)	Laden des Grafikpakets
plot(f(x),x=a..b)	graphische Darstellung der Funktion f(x) im Intervall [a,b] ("2D-Plot")
implicitplot(f(x,y)=0,x=a..b,y=c..d)	grafische Darstellung der Lösungsmenge von f(x,y)=0 im Rechteck [a..b,c..d]
plot([f1(x),f2(x)],x=a..b)	Gleichzeitige grafische Darstellung der Funktionen f1(x) und f2(x) im Intervall [a,b]
p1:=plot(f1(x),x=a..b):	
p2:=plot(f2(x),x=a..b):	
display([p1,p2]);	Gleichzeitige grafische Darstellung der Funktionen f1(x) und f2(x) im Intervall [a,b]
plot([x(t),y(t),t=a..b])	graphische Darstellung einer in Parameterdarstellung betrachteten Funktion f(t)=f(x(t),y(t)) im Intervall t=a..b
plot3d(f(x,y),x=a..b,y=c..d)	grafische Darstellung der Funktion f(x,y) über dem Rechteck [a..b,c..d] ("3D-Plot")

Beispiele:

```
> # Darstellung von mehreren Graphen:
> y1:=10*x^3-80*x^2+190*x-120;
```

$$y1 := 10\,x^3 - 80\,x^2 + 190\,x - 120$$

```
> solve(28*x+4*y2=231,y2);
```

$$-7\,x + \frac{231}{4}$$

```
> y2:=%;
```

$$y2 := -7\,x + \frac{231}{4}$$

```
> solve(6*x-4*y3-75=0,y3);
```

$$\frac{3\,x}{2} - \frac{75}{4}$$

```
> y3:=%;
```

$$y3 := \frac{3\,x}{2} - \frac{75}{4}$$

```
> with(plots):
Warning, the name changecoords has been redefined

> plot({(y1,x=1..4.5),(y2,x=4.5..9),(y3,x=9..15)},x=1..15); # geht
nicht
Warning, unable to evaluate 3 of the 6 functions to numeric values in the region;
see the plotting command's help page to ensure the calling sequence is correct

> a:=plot(y1,x=1..4.5):
> b:=plot(y2,x=4.5..9):
> c:=plot(y3,x=9..15):
> display([a,b,c]);
```




```
> plot({y1,diff(y1,x),diff(y1,x,x)},x=1..4.5);
```




```
> # Nullstellen:
> solve(y1=0,x);
```

$$1, 3, 4$$

```
> solve(diff(y1,x)=0,x);
```

$$\frac{8}{3}+\frac{\sqrt{7}}{3}, \frac{8}{3}-\frac{\sqrt{7}}{3}$$

```
> solve(diff(y1,x,x)=0,x);
```

$$\frac{8}{3}$$

```
>
>
> # Zeichnungen von Funktionen zweier Variablen
> x:=evaln(x);
```

$$x := x$$

```
> y:=evaln(y);
```

$$y := y$$

```
> f:=(x,y)->x^2+y^2-2*y;
```

$$f := (x, y) \to x^2 + y^2 - 2\,y$$

```
> 'f(x,y)'=f(x,y);
```

$$\mathrm{f}(x, y) = x^2 + y^2 - 2\,y$$

```
> plot3d(f(x,y),x=-5..5,y=-5..5,axes=BOXED);
```




```
> plots[contourplot](f(x,y),x=-5..5,y=-5..5,scaling=CONSTRAINED); #
Darstellung von Neveaulinien
```




```
>
> x:=evaln(x);
```

$$x := x$$

```
> y:=evaln(y);
```

$$y := y$$

```
> f:=(x,y)->sqrt(y-x^2);
```

$$f := (x, y) \to \sqrt{y - x^2}$$

```
> 'f(x,y)'=f(x,y);
```

$$\mathrm{f}(x, y) = \sqrt{y - x^2}$$

```
> plot3d(f(x,y), x=-5..5, y=-1..9, axes=BOXED, numpoints=50^2);
```




```
> plots[contourplot](f(x,y), x=-5..5, y=-1..9,
numpoints=100^2,scaling=CONSTRAINED, view=[-5..5,-1..9]);
```